

SERVICE

par André Chéramy

Présentation

Quoi de plus agaçant qu'une ligne service polluée par l'utilisation d'un programme et qui affiche avec obstination un message, une en tête ou un copyright qui n'ont plus cours... Quoi de plus agaçant qu'un CAPS qui vient défigurer la page écran que vous venez de composer et qui reste hors d'atteinte du curseur, des PRINT@ et autres PLOT...

Bien sûr, il y a des trucs du genre FOR I=#BB80 TO #BBA7:POKE I,32:NEXT pour nettoyer la ligne service ou encore POKE #BBA3,0 pour effacer CAPS ou encore POKE#20C,#7F et POKE#20C,#FF pour passer en mode «minuscule» ou «MAJUSCULE» sans afficher le très pratique, mais affreux «CAPS».

Mais si vous voulez personnaliser votre écran courant avec un titre ronflant multicolore et clignotant, les POKES, ça devient pénible. Ou encore, si vous voulez éditer la ligne service d'une page écran en mode TEXT, sans polluer cette page elle même... Alors il y a SERVICE qui fait tout cela pour vous, très simplement!

Package:

Seulement deux fichiers: SERVICE.COM qui est le programme proprement dit et SERVICE.HLP qui est l'écran d'aide. Les amateurs de listing pourront s'en donner à coeur joie. Les paresseux devront attendre la prochaine disquette trimestrielle...

Mode d'emploi

Lancer SERVICE. Un menu s'affiche, vous pouvez:

A > Afficher un écran d'Aide.

C > Modifier la ligne service de l'écran Courant (c'est le choix par défaut).

D > Modifier la ligne service d'un écran TEXT, présent sur Disquette. Le nom du fichier correspondant vous sera demandé. Les modifications que vous ferez se feront non seulement sur cet écran TEXT, mais aussi sur la ligne service de l'écran courant.

P > Modifier seulement la couleur du fond (PAPER) (concerne la première case de la ligne service). Cette modification pourra se faire automatiquement (en recopiant l'attribut présent dans la première case de la ligne suivante) ou manuellement (en indiquant la couleur désirée).

I > Modifier seulement la couleur de l'encre (INK) (concerne la deuxième case de la ligne service). Cette modification pourra se faire automatiquement (en recopiant l'attribut présent dans la deuxième case de la ligne suivante) ou manuellement (en indiquant la couleur désirée).

R > Effacer la ligne service, en la Remplissant avec des espaces (sauf les deux premières cases, afin de ne pas altérer les attributs PAPER et INK).


M > Afficher un Message sur la ligne service. Ce message peut inclure des attributs vidéo (de @ à G pour la couleur INK; de P à W pour la couleur PAPER et les attributs H, I, L, et M (mais pas J, K, N et O qui impliquent la double hauteur). Pour taper votre message, vous pouvez:

- Utiliser les flèches gauche et droites.

- Effacer le caractère situé à gauche du curseur (représenté par une flèche affichée à la ligne en dessous) avec la touche DEL.

- Basculer entre les modes «MAJUSCULE» et «minuscule» à l'aide du classique CTRL/T. L'habituel «CAPS» sur la ligne service est remplacé par «MAJ» ou «min» sur la ligne en dessous.

- Placer des attributs vidéo en appuyant sur ESC puis sur une lettre de @ à W (sauf J, K, N, et O). Cette option marche même si vous êtes en mode «minuscule».



- Entrer un caractère spécial non représenté au clavier en appuyant sur CTRL/A puis en tapant le code ASCII de ce caractère (32 à 127 pour le jeu standard mais attention seulement 32 à 95 pour le mode semi graphique!).

- Accéder directement au caractère ê du Sedoric (ACCENT SET) en pressant simultanément FUNCT et ^

- Faire RETURN pour terminer.

F > Pour Finir, c'est à dire sauver votre fichier d'écran TEXT si vous en aviez un en cours, puis sortir de SERVICE et retourner au Ready. NB: l'ancien fichier sera conservé avec l'extension «.BAK».

Le listing

```
1000 '*** SERVICE A.Ch{ramy 1991 ***
1010 'V26019111131
1020 'Menu
1030 HIMEM#7000:E$="C":MOVE#BB80,#BBA7,#7000:'Ce move sert @ pr{charger
1040 CLS:PLOT37,1,"MAJ":POKE#20C,#FF:PRINT:PRINT:PRINT:PRINT:PRINT
1050 PRINT"Ce programme permet d'éditer la ligne"
1060 PRINT"service en haut de l'cran TEXT."
1070 PRINT"-----":PRINT
1080 PRINT:PRINT"Afficher un {cran d'Aide? (A)"
1090 PRINT:PRINT"Ecran courant ou sur disquette? (C/D)"
1100 PRINT:PRINT"Editer INK ou PAPER? (I/P)"
1110 PRINT:PRINT"Remplir avec des espaces? (R)"
1120 PRINT:PRINT"Afficher un message? (M)"
1130 PRINT:PRINT"Finir et retourner au Ready? (F)"
1140 PRINT:PRINT"Votre choix?";:GET R$
1150 IF R$="A" THEN !SERVICE.HLP:GET R$:GOTO 1040
1160 IF R$="C" THEN E$="C":GOTO 1040
1170 IF R$="D" THEN GOSUB 1250:GOTO 1040
1180 IF R$="I" THEN GOSUB 1310:GOTO 1040
1190 IF R$="P" THEN GOSUB 1430:GOTO 1040
1200 IF R$="R" THEN GOSUB 1550:GOTO 1040
1210 IF R$="M" THEN GOSUB 1580:GOTO 1040
1220 IF R$="F" THEN GOTO 2060
1230 PING:GOTO 1040
1240 'S/P Page-{cran sur disquette"
1250 CLS:PRINT:PRINT:INPUT"Donnez le nom du fichier";F$;!LOAD F$:E$="D"
1260 MOVE#BB80,#BFDF,#7B80
1270 GET A$:CLS:PRINT:PRINT:PRINT"Etait-ce le bon? (O/N)";:GET R$
1280 IF R$="N" THEN 1250
1290 IF R$<>"O" THEN PING:GOTO 1270 ELSE RETURN
1300 'S/P Edition de Ink
1310 CLS:PRINT:PRINT:PRINT"Modification auto ou manuelle? (A/M)";:GET R$
1320 IF E$="D" THEN IK=PEEK(#7BA9) ELSE IK=PEEK(#BBA9)
1330 IF R$="A" THEN POKE#BB81,IK:POKE#7B81,IK:GOTO 1400
1340 IF R$<>"M" THEN PING:GOTO 1310
1350 PRINT:PRINT:PRINT"Noir=0, Rouge=1, Vert=2, Jaune=3,"
1360 PRINT"Bleu=4, Magen=5, Cyan=6, Blanc=7,"
1370 PRINT:PRINT"Votre Choix?";:GET R$
1380 IF R$<"0" OR R$>"7" THEN PING:GOTO 1370
1390 POKE#7B81,VAL(R$):POKE#BB81,VAL(R$)
1400 PRINT:PRINT"OK? (O/N)";:GET R$
1410 IF R$<>"O" THEN 1310 ELSE RETURN
1420 'S/P Edition de Paper
1430 CLS:PRINT:PRINT:PRINT"Modification auto ou manuelle? (A/M)";:GET R$
1440 IF E$="D" THEN PP=PEEK(#7BA8) ELSE PP=PEEK(#BBA8)
1450 IF R$="A" THEN POKE#BB80,PP:POKE#7B80,PP:GOTO 1520
1460 IF R$<>"M" THEN PING:GOTO 1430
1470 PRINT:PRINT:PRINT"Noir=0, Rouge=1, Vert=2, Jaune=3,"
1480 PRINT"Bleu=4, Magen=5, Cyan=6, Blanc=7,"
1490 PRINT:PRINT"Votre Choix?";:GET R$
1500 IF R$<"0" OR R$>"7" THEN PING:GOTO 1490
```



```

1510 POKE#7B80,VAL(R$)+16:POKE#BB80,VAL(R$)+16
1520 PRINT:PRINT"OK? (O/N)";:GET R$
1530 IF R$<>"O" THEN 1430 ELSE RETURN
1540 'S/P Remplir avec des espaces
1550 FOR I=#7B80 TO #7BA7:POKEI,32:NEXT
1560 FOR I=#BB80 TO #BBA7:POKEI,32:NEXT:RETURN
1570 'S/P Afficher un message
1580 CLS:PRINT:PRINT:PRINT"Quel est ce message?":PRINT
1585 NC=0:'Nombre de caract}res (0 @ 39) et pointeur sur ligne service
1590 PRINT"(Touches DEL, -> et <- utilisables)"
1600 PRINT"(CTRL/A Entr{e directe code ASCII)"
1610 PRINT"(FUNCT/^ pour e circonflexe)"
1620 PRINT"(ESC puis Attribut si besoin)"
1630 PRINT"(CTRL/T bascule min/MAJ)"
1640 PRINT"(RETURN pour finir.)":PRINT:PRINT"?";:GOSUB 1900:PLOT37,1,"MAJ"
1650 'Saisie de chaine avec attributs
1670 GET A$:A=ASC(A$):FS=PEEK(#209)
1680 IF A=1 THEN FS=#38:INPUT A:'CTRL/@ Entr{e ASCII
1690 IF A=8 THEN 1920:'Gauche
1700 IF A=9 THEN 1950:'Droite
1710 IF A=13 THEN RETURN:'C est fini
1720 IF A=20 THEN 1980:'CTRL/T Bascule min/MAJ
1730 IF A=27 THEN 1800:'Entr{e Attribut
1740 IF A=127 THEN 2010:'DEL
1750 IF A<32 OR A>127 THEN PING:GOTO 1670:'Reste seulement les car affichables
1760 IF (A=94 OR A=54) AND (FS=#A5 OR FS=#A7) THEN A=126:' e circonflexe
1770 'Cas g{n{ral, caract}res normaux
1780 GOSUB 1880:NC=NC+1:IF NC>39 THEN NC=0
1790 GOSUB 1900:GOTO 1670
1800 'Entr{e Attribut
1810 GET A$:A=ASC(A$)
1820 IF A>96 AND A<120 THEN A=A-32
1830 IF A<64 OR A>87 THEN ZAP:GOTO 1810:' @ - W seulement
1840 A=A-64:GOSUB 1880:'On poke l'attribut correspondant
1850 NC=NC+1:IF NC>39 THEN NC=0:'Position Next Caract}re
1860 GOSUB 1900:GOTO 1670:'Fl}che curseur et rebouclage saisie suivante
1870 'Double poke du caract}re ou de l'attribut
1880 POKE#7B80+NC,A:POKE#BB80+NC,A:RETURN
1890 'Affichage fl}che curseur
1900 PLOT0,0," "
1910 POKE#BBA8+NC,94:RETURN
1920 'Fl}che Gauche
1930 NC=NC-1:IF NC<0 THEN NC=39:'Position Next Caract}re
1940 GOSUB 1900:GOTO 1670:'Fl}che curseur et rebouclage saisie suivante
1950 'Fl}che Droite
1960 NC=NC+1:IF NC>39 THEN NC=0:'Position Next Caract}re
1970 GOSUB 1900:GOTO 1670:'Fl}che curseur et rebouclage saisie suivante
1980 'CTRL/T Bascule min/MAJ
1990 IF PEEK(#20C)=#FF THEN POKE#20C,#7F:PLOT37,1,"min":GOTO 1670
2000 POKE#20C,#FF:PLOT37,1,"MAJ":GOTO 1670
2010 'Del
2020 NC=NC-1:IF NC<0 THEN NC=39:'Position Next Caract}re
2030 A=32:GOSUB1880:'Remplace caract}re de gauche par espace
2040 GOSUB1900:GOTO1670:'Fl}che curseur meme position et rebouclage saisie
2050 'S/P Finir et retourner au Ready
2060 IF E$="D" THEN MOVE#7B80,#7FDF,#BB80:GET R$:SAVEU F$,A#BB80,E#BFDF
2070 CLS:END

```



Comme vous pouvez le constater ce programme inédit date de 1991.
 Vous aussi fouillez vos archives et faites profiter les copains de vos oeuvres !